
script DFLT {

  feature ContextualAlternates1;

}

class @lower [a b c d e f g h i j k l m n o p q r s t u v w x y z e_low_out e_mid_in

 e_mid_in_low_out o_high_in o_low_in o_high_out o_high_in_high_out o_low_in_high_out

 o_mid_out o_high_in_mid_out o_low_in_mid_out e_low_in e_low_in_low_out g_high_in

 g_low_in g_low_out g_high_in_low_out g_low_in_low_out];

class @not.e [a b c d f g h i j k l m n o p q r s t u v w x y z o_high_in o_low_in

 o_high_out o_high_in_high_out o_low_in_high_out o_mid_out o_high_in_mid_out

 o_low_in_mid_out g_high_in g_low_in g_low_out g_high_in_low_out g_low_in_low_out];

class @high.out [o_high_out o_high_in_high_out o_low_in_high_out];

class @mid.out [o_mid_out o_high_in_mid_out o_low_in_mid_out];

class @e [e e_low_out e_mid_in e_mid_in_low_out e_low_in e_low_in_low_out];

class @g [g g_high_in g_low_in g_low_out g_high_in_low_out g_low_in_low_out]; 

class @low.out [e_low_out e_low_in_low_out e_mid_in_low_out g_low_out g_low_in_low_out

 g_high_in_low_out];

feature ContextualAlternates1 calt {

  lookup class.1.high.out;

  lookup class.1.mid.out;

  lookup class.2.low.out;

  lookup high.out;

  lookup low.out;

  lookup oe;

}

lookup class.1.high.out {

  context @lower (@not.e);

  sub 0 class.1.high.out.1;

}

lookup class.1.high.out.1 {

  sub o -> o_high_out;

  sub o_high_in -> o_high_in_high_out;

  sub o_low_in -> o_low_in_high_out;

}

lookup class.1.mid.out {

  context @lower (@e);

  sub 0 class.1.mid.out.1;

}

lookup class.1.mid.out.1 {

  sub o -> o_mid_out;

  sub o_high_in -> o_high_in_mid_out;

  sub o_low_in -> o_low_in_mid_out;

}

lookup class.2.low.out {



  context @lower (@lower);

  sub 0 class.2.low.out.1;

}

lookup class.2.low.out.1 {

  sub e -> e_low_out;

  sub e_mid_in -> e_mid_in_low_out;

  sub e_low_in -> e_low_in_low_out;

  sub g -> g_low_out;

  sub g_high_in -> g_high_in_low_out;

  sub g_low_in -> g_low_in_low_out;

}

lookup high.out {

  context (@high.out) @lower;

  sub 0 high.out.1;

}

lookup high.out.1 {

  sub o -> o_high_in;

  sub o_high_out -> o_high_in_high_out;

  sub o_mid_out -> o_high_in_mid_out;

  sub g -> g_high_in;

  sub g_low_out -> g_high_in_low_out;

}

lookup low.out {

  context (@low.out) @lower;

  sub 0 low.out.1;

}

lookup low.out.1 {

  sub o -> o_low_in;

  sub o_high_out -> o_low_in_high_out;

  sub o_mid_out -> o_low_in_mid_out;

  sub e -> e_low_in;

  sub e_low_out -> e_low_in_low_out;

  sub g -> g_low_in;

  sub g_low_out -> g_low_in_low_out;

}

lookup oe {

  context (@mid.out) @lower;

  sub 0 oe1;

}

lookup oe1 {

  sub e -> e_mid_in;

  sub e_low_out -> e_mid_in_low_out;

}


